In[1]:=
```
wedgeRules = {
   Wedge[a___, dx[i_], b___, dx[i_], c___] → 0,
   Wedge[a___, dx[i_], dx[j_], c___] /; j < i → -Wedge[a, dx[j], dx[i], c], (* Alternation Rule *)

   a_ ∧ forms_Plus :↦ (a ∧ # &) /@ forms, forms_Plus ∧ a_ :↦ (# ∧ a &) /@ forms,
   (* Distributive Rule *)

   Times[a_, dx[i_]] ∧ b_ → a (dx[i] ∧ b),
   b_ ∧ Times[a_, dx[i_]] → a (b ∧ dx[i]), Times[a_, dx[i_] ∧ dx[j_]] ∧ b_ → a (dx[i] ∧ dx[j] ∧ b),
   b_ ∧ Times[a_, dx[i_] ∧ dx[j_]] → a (b ∧ dx[i] ∧ dx[j])
   (* Scalars "commute" *)
  };

(* Exterior derivative of 1-forms and 2-forms in ℝ^3 *)
```

In[2]:=
```
d[forms_Plus] := d[#] & /@ forms
d[Times[f_, dx[i_]]] := Sum[D[f, x_k] dx[k] ∧ dx[i], {k, 3}]
d[Times[f_, dx[i_] ∧ dx[j_]]] := Sum[D[f, x_k] dx[k] ∧ dx[i] ∧ dx[j], {k, 3}]
```

In[5]:=
```
(* Formatting hacks. *)
fr = {Derivative[i_, j_, k_][f_][___] → "∂"^(i+j+k) f / ("∂x_1"^i "∂x_2"^j "∂x_3"^k)}; fr2 = {f_[x_1, x_2, x_3] → f};
format[expr_] := expr // Replace[#, fr, 5] & // Replace[#, fr2, 3] &
```

In[7]:=
```
(* Define a pair of 1-forms in terms of Euclidean coordinate functions. *)
ϕ = Sum[f_k[x_1, x_2, x_3] dx[k], {k, 3}];
ψ = Sum[g_k[x_1, x_2, x_3] dx[k], {k, 3}];

(* Verify Theorem 6.4 (3). *)
```

In[9]:=
```
d[ϕ ∧ ψ] //. wedgeRules // FullSimplify // format[#] &
```

Out[9]=
$$\left(\left(-\frac{\partial f_2}{\partial x_3}+\frac{\partial f_3}{\partial x_2}\right) g_1+\left(\frac{\partial f_1}{\partial x_3}-\frac{\partial f_3}{\partial x_1}\right) g_2+f_3 \left(\frac{\partial g_1}{\partial x_2}-\frac{\partial g_2}{\partial x_1}\right)+\left(-\frac{\partial f_1}{\partial x_2}+\frac{\partial f_2}{\partial x_1}\right) g_3+f_2 \left(-\frac{\partial g_1}{\partial x_3}+\frac{\partial g_3}{\partial x_1}\right)+f_1 \left(\frac{\partial g_2}{\partial x_3}-\frac{\partial g_3}{\partial x_2}\right)\right) dx[1] \wedge dx[2] \wedge dx[3]$$

In[10]:=
```
d[ϕ] ∧ ψ - ϕ ∧ d[ψ] //. wedgeRules // FullSimplify // format[#] &
```

Out[10]=
$$\left(\left(-\frac{\partial f_2}{\partial x_3}+\frac{\partial f_3}{\partial x_2}\right) g_1+\left(\frac{\partial f_1}{\partial x_3}-\frac{\partial f_3}{\partial x_1}\right) g_2+f_3 \left(\frac{\partial g_1}{\partial x_2}-\frac{\partial g_2}{\partial x_1}\right)+\left(-\frac{\partial f_1}{\partial x_2}+\frac{\partial f_2}{\partial x_1}\right) g_3+f_2 \left(-\frac{\partial g_1}{\partial x_3}+\frac{\partial g_3}{\partial x_1}\right)+f_1 \left(\frac{\partial g_2}{\partial x_3}-\frac{\partial g_3}{\partial x_2}\right)\right) dx[1] \wedge dx[2] \wedge dx[3]$$

In[11]:=
```
(* Or just... *)
d[ϕ ∧ ψ] - (d[ϕ] ∧ ψ - ϕ ∧ d[ψ]) //. wedgeRules // FullSimplify
```

Out[11]=
```
0
```